

# SAS<sup>®</sup> 9 ODS Layout Tip Sheet

## Absolute Layout Example

```
options nocenter obs=5;
ods pdf file='absolute.pdf';

ods layout absolute
  x=0cm y=0cm
  width=17cm height=23cm
  style={background=lightgray};

ods region x=1cm y=1cm
  width=10cm height=10cm
  style={background=lightred};

proc report data=sashelp.class
  nowd;
  column name age;
run;

ods region x=2cm y=2cm
  width=10cm height=10cm
  style={background=lightblue};

proc report data=sashelp.cars
  nowd;
  column make model;
run;

ods region x=3cm y=3cm
  width=10cm height=10cm
  style={background=lightgreen};

proc report data=sashelp.shoes
  nowd;
  column product sales;
run;

ods layout end;
ods pdf close;
```

Name	Age
Alfred	14
Alice	13
Barbara	13
Carol	14
Henry	14

Make	Model	Total Sales
Acura	Accord	\$29,761
Acura	CR-V	\$67,242
Acura	Merri Casual	\$76,793
Acura	Merri Dress	\$62,819
Acura	Sliper	\$88,641

## Gridded Layout Example

```
ods pdf file='gridded.pdf';

ods layout gridded
  columns=2 width=5in
  column gutter=.25in
  style={background=lightgray};

ods region
  style={background=lightblue};

proc means data=sashelp.class mean;
run;

proc report nowd
  data=sashelp.class (obs=5)
  nowd;
run;

ods region width=2in
  style={background=lightblue};

proc print
  data=sashelp.class (obs=5
  keep=name age);
run;

ods layout end;
ods pdf close;
```

Variable	Mean
Age	13.3157895
Height	62.3368421
Weight	100.0263158

Obs	Name	Age
1	Alfred	14
2	Alice	13
3	Barbara	13
4	Carol	14
5	Henry	14

Name	Sex	Age	Height	Weight
Alfred	M	14	69	112.5
Alice	F	13	56.5	84
Barbara	F	13	65.3	98
Carol	F	14	62.8	102.5
Henry	M	14	63.5	102.5



# ODS Layout Tip Sheet

This tip sheet places frequently used information in one place, on one sheet of paper, so you don't have to search through the online documentation. It also gives you something to take home, type in, and try.

Creating custom reports that show your information in just the right way is now possible using ODS Layout. ODS Layout has two distinct forms, absolute and gridded. Each form is suited to different reporting tasks. The most important decision when using ODS Layout is selecting the type that best suits your report design. Absolute layout is valid for Printer destinations only. Gridded layout is valid for Printer, HTML, and SASReport.



SAS Institute Inc. World Headquarters  
+1 919 677 8000 [sas.com/offices](http://sas.com/offices)

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies. Copyright © 2013, SAS Institute Inc. All rights reserved. 672768\_02.13

# SAS<sup>®</sup>9 ODS Layout Tip Sheet

## Common Layout Options and Values

The `style=` option and option values listed below are used by the layout and region options in this tip sheet.

### ***dimension***

indicates a positive integer or floating point value followed by one of the following units:

- in** – inches
- cm** – centimeters (2.54 cm = 1 inch)
- mm** – millimeters (25.4 mm = 1 inch)
- % (pct)** - percentage (percentage of container width or height for `width=` and `height=`)
- px** – pixels
- pt** – points (1/72 inch)

### ***integer***

indicates a positive integer.

### **`style=style-element {style-attributes}`**

all layout and region statements accept the `style=` option. This statement allows you to specify a style element and style overrides.

*(style= is pre-production in SAS 9.4)*

## Absolute Layout Syntax

```
ods layout absolute options;
```

### ***x=dimension***

specifies horizontal position of the layout, which extends to the right. If omitted, it defaults to 0.

### ***y=dimension***

specifies vertical position of the layout, which extends down. If omitted, it defaults to the current vertical position on the page.

## Absolute Layout Syntax

*(continued)*

### ***width=dimension***

specifies horizontal width of the layout. If omitted, it defaults to the maximum horizontal space available on the page.

### ***height=dimension***

specifies vertical height of the layout. If omitted, it defaults to the maximum vertical space available on the page.

## Absolute Layout Region Syntax

```
ods region options;
```

### ***x=dimension***

specifies horizontal position of the region within the layout, which extends to the right. If omitted, it defaults to 0.

### ***y=dimension***

specifies vertical position of the region within the layout, which extends down. If omitted, it defaults to 0.

### ***width=dimension***

specifies horizontal width of the region, and is restricted by the layout dimensions. If omitted, it defaults to the maximum horizontal space available in the layout container.

### ***height=dimension***

specifies vertical height of the region and is restricted by the layout dimensions. If omitted, it defaults to the maximum vertical space available in the layout container.

## Gridded Layout Syntax

```
ods layout gridded options;
```

### ***x=dimension***

specifies horizontal position of the layout, which extends to the right of this position. If omitted, the layout container is centered on the page.

### ***y=dimension***

specifies vertical position of the layout, which extends down from this position. If omitted, it defaults to the current vertical position on the page.

### ***width=dimension***

specifies horizontal width of the layout. If omitted, it defaults to the maximum horizontal space needed to display all regions.

### ***height=dimension***

specifies vertical height of the layout. If omitted, it defaults to the maximum vertical space needed to display all regions.

### ***columns=integer***

specifies fixed number of columns in the gridded layout. If omitted, it defaults to 1.

### ***column\_widths=(dimension-1 ... dimension-n)***

specifies width of each column.

### ***column\_gutter=dimension***

specifies space between each column.

### ***rows=integer***

specifies fixed number of rows in the layout. If omitted, the number of regions created in the vertical direction is used.

### ***row\_heights=(dimension-1 ... dimension-n)***

specifies height of each row.

## Gridded Layout Syntax

*(continued)*

### ***row\_gutter=dimension***

specifies space between each row.

### ***order\_type=row\_major | column\_major***

indicates which direction the grid is populated. *Printer destination only.*

### ***advance=explicit | proc | table | bygroup***

specifies which output type advances to the next region.

## Gridded Layout Region Syntax

```
ods region options;
```

### ***width=dimension***

specifies horizontal width of the region and is restricted by the layout dimensions.

### ***height=dimension***

specifies vertical height of the region and is restricted by the layout dimensions.

### ***column=integer***

specifies the current column position in the gridded layout.

### ***column\_span=integer***

specifies the number of columns that the region will occupy. The default value is 1.

### ***row=integer***

specifies the current row position in the gridded layout.

### ***row\_span=integer***

specifies the number of rows that the region will occupy. The default value is 1.